

# AVEVA™



## PML基础

王元  
AVEVA中国  
2005.11.19



## 如何学习PML编程



- ▼ 需要具备的基本知识
  - 熟悉Pdms的基本操作
  - 了解Pdms的运行机制，模块之间的关系
  - Pdms属性的操作
  - Pdms命令的操作
  - Pml语法(本教程)
  - 通过Pdms菜单查找示范程序

VANTAGE Plant Design

## 课程将包括...



- ▼ PML介绍
  - PML概念
  - PML功能
  - PML构成
- ▼ PML程序语言
  - 变量的定义及用法
  - 循环Do、判断If、跳转Golabel
  - 错误处理
  - 文件处理
  - 函数Functions、方法Method
  - 对话框Forms和菜单Menu

VANTAGE Plant Design

## 课程结束后 ...



### ▼ 您会具有以下知识 ...

- PML构成及功能
- 编写简单的宏
- PML变量的定义及应用
- PML函数/方法的定义及调用
- PML窗体/对象的定义及调用

VANTAGE Plant Design

## PML - 功能强大的语言



- ▼ **P**rogrammable **M**acro **L**anguage 可编程宏语言
- ▼ PML2 基于面向对象(Object Oriented)概念的编程语言
- ▼ 支持用户自定义对象类型(Object Types)
- ▼ 简单易学，与PDMS无缝连接
- ▼ 丰富的内置函数，方法及对象
- ▼ 最简单的对话框，菜单编写语言

VANTAGE Plant Design

## PML构成



### ▼ Macros

- Macros是包含PDMS 命令序列的ASCII文件
- Macros 在 PDMS 中通过 `$m /FILENAME`来执行

### ▼ PML

- 变量(Variables)
- 判断语句(if Construct)
- 循环(Do loops)
- 错误处理(Error Handling)
- 文件处理(Files and Directories)

### ▼ 窗体和菜单

- PDMS大多数应用程序都由对话框 (Forms)和菜单 (Menus)来驱动

VANTAGE Plant Design

## 一个简单的 Macro



### ▼ 宏是包含PDMS 命令序列的一个文本文件

### ▼ 运行宏

`$M /MyFile`

ASCII 文件  
MYFILE

```
NEW EQUIP /FRED  
NEW BOX  
XLEN 300 YLEN 400 ZLEN 600  
NEW CYL DIA 400 HEI 600  
CONN P1 TO P2 OF PREV
```

VANTAGE Plant Design

## 参数化宏Parameterized Macro



- ▼ 宏可以参数化
- ▼ 文件名后面的字符串是参数(parameters)  
`$M /MyFile NEWEQUIP 300 400 600`

ASCII 文件  
MYFILE

```
NEW EQUIP /$1  
NEW BOX  
XLEN $2 YLEN $3 ZLEN $4  
NEW CYL DIA $3 HEI $4  
CONN P1 TO P2 OF PREV
```

VANTAGE Plant Design

## 宏的参数



- ▼ Macros 可以有多达 9 个由空格分隔的参数.  
`$M /MyFile PML TRAINING 55 66`
- ▼ 文本字符串可以以单个参数输入  
`$M /MyFile $< PML TRAINING $> 55 66`
- ▼ `$<` 和 `$>` 是分隔符，在它们之间的任何字符都被认为是单个参数

VANTAGE Plant Design

## 变量Variables



### ▼ 变量基本概念

- 变量用于存储数值,变量必须有名字,变量的数值可以改变,但变量名是固定的.

### ▼ PML变量

- 在PML2中变量是一个对象(Objects)
  - 每一个对象(变量)有一个唯一的名字
  - 每一个对象都有一套函数与之相关联,这些函数称为方法(Methods).方法用于处理对象中的数据
  - 生成对象时必须指明对象类型(Object type),不同的对象类型对应不同的方法

VANTAGE Plant Design

## 对象(变量)类型Object Types



### ▼ 内置(Build-in)的对象类型

- 字符串类型(String).如'Hello World'
- 实数类型(Real).整数类型包括在实数类型中,如99
- 布尔类型(Boolean).用于逻辑表达式,如True,False
- 数组类型(Array).可以存储任意类型的数据

### ▼ 系统定义(System-defined)的对象类型

- 指在PDMS中的变量类型,如Positon,Reference

### ▼ 用户自定义(User-defined)的对象类型

VANTAGE Plant Design

## 生成、查询、删除变量



- ▼ 使用命令行或者文本文件测试

- ▼ 通过赋值声明变量类型

```
!MyString = 'Hello World'
```

```
Q var !MyString
```

```
显示<STRING> 'Hello World'
```

- ▼ 直接声明变量类型

```
!Length = Real()
```

```
Q Var !Length
```

```
显示<REAL> Unset
```

- ▼ 删除变量用到方法

```
!MyString.Delete()
```

VANTAGE Plant Design

## 变量命名规则(Naming Conventions)



- ▼ 局部变量(Local)和全局变量(Global)

```
!SurfaceArea          ! 表示局部变量
```

```
!!Area                !!表示全局变量
```

- ▼ 变量名最长 16 个字符(不包括!和!!)，变量名可以包含字母和数字

- ▼ 变量名不能用数字和点(.)开头

- ▼ 变量名的大小写不敏感

- ▼ 建议:一个变量一个用途,最好给变量名一个有意义的名字,并且区分大小写

```
!!StartInUpperCase
```

VANTAGE Plant Design

## PDMS属性类型-系统定义变量类型



- ▼ 名字(Name) 如Name
- ▼ 字符串(String) 如Description,Function
- ▼ 实数(Real) 如Angle,Temperature,Rating
- ▼ 布尔(Boolean) 如Lock,Shop,Built
- ▼ 数组(Array) 如Level
- ▼ 参考(Reference) 如Spref,Catref,Lstube,Ptref,Gmref
- ▼ 关键字(Word) 如Type,Purpose
- ▼ 位置(Position) 如Position,Hposition
- ▼ 方位(Orientation) 如Orientation
- ▼ 方向(Direction) 如HDirection

VANTAGE Plant Design

## !!ce



- ▼ 特殊全局变量!!CE
  - 得到当前元素的参考，DBref类型
  - ‘.’点操作符可以提取元素的属性和隐含属性

VANTAGE Plant Design



## 变量提取属性



- ▼ 用变量提取属性值,变量名=DBref变量.属性名

```
!Name = !!ce.Name
!Desc = !!ce.Description
!Pspec = !!ce.Pspec
!Temp = !!ce.Temp
!Purp = !!ce.Purp
!Rating = !!ce.cref.pspec.rating
!Pos = !!ce.Pposition[3]      $*P-point[3]坐标
!Dir = !!ce.Pdirection[1]     $*P-point[1]方向
```

- ▼ 查询属性类型

```
!Ref = Ref
Q var !Ref
```

VANTAGE Plant Design

## 变量给PDMS属性赋值



- ▼ 属性名 = 变量名
  - 名字赋值,变量中第一个字符必须是/

```
!!ce.Name = !Name
!!ce.Describsion = !Desc
!!ce.Temp = !Temp
!!ce.Pspec = !Pspec
!!ce.Purp = !Purp
```

VANTAGE Plant Design

## 常用的赋值方法



### ▼ 可以不是当前元素

!A = !!CE

!!CE = !!CE.Owner

!A.Built = TRUE

### ▼ 坐标的变化

!Pos = !!CE.Position

!Pos.Up = 2000

!!CE.Position = !Pos

VANTAGE Plant Design

## 练习-变量给属性赋值



### ▼ 使用文本编辑器编辑

- 获取当前元素的Name和Type
- 新建同样类型的元素
- 在命名中追加-NEW

VANTAGE Plant Design

## PDMS命令中使用变量



### ▼ 变量前加\$表示提取变量的值

```
!type = !!ce.type  
!name = !!ce.name  
!newname = !name + '-NEW'  
New $!type $!newname
```

### ▼ 直接组合字符串

```
New $!type $!name-NEW
```

VANTAGE Plant Design

## 方法(Method)和函数(Function)



### ▼ 函数(Function)是执行特定功能的子程序

### ▼ 方法(Method)是对象(变量)的函数

- 在PML2中变量是一个对象(Objects)
  - 每一个对象(变量)有一个唯一的名字
  - 每一个对象都有一套函数与之相关联,这些函数称为方法(Methods).方法用于处理对象中的数据
  - 生成对象时必须指明对象类型(Object type),不同的对象类型对应不同的方法
- 方法并不改变对象的类型和对象中的值

### ▼ 参考手册*Cadcentre Software Customisation Reference Manual*

VANTAGE Plant Design

## 方法-字符串对象(String Object)



|            |                                           |
|------------|-------------------------------------------|
| ▼ 字符串长度    | Length()                                  |
| ▼ 类型转换     | Real(),Position(),Boolean()               |
| ▼ 大小写转换    | LowCase(),UpCase()                        |
| ▼ 两端截取     | After(str2),Before(str2),Substring(index) |
| ▼ 中间截取     | Substring(index,nchars)                   |
| ▼ 分隔符截取    | Part(nth),Part(nth,delim)                 |
| ▼ 替换       | Replace(str2,str3)                        |
| ▼ 匹配       | Match(str2),MatchWild(str2)               |
| ▼ 分割(返回数组) | Split()                                   |
| ▼ 修剪       | Trim(),Trim(option)                       |

VANTAGE Plant Design

## 方法实例



### ▼ 方法示例

```
!str = 'hello how are you!'
!newstr = !str.after('hello').trim().upcase()
q var !newstr
<String> 'HOW ARE YOU!'
```

### ▼ 练习-将字符串改为'where are you?'

```
!str = 'hello how are you!'
!newstr = !str.after('hello').trim().replace('how', 'where').replace('!', '?')
```

VANTAGE Plant Design

## PML表达式



### ▼ 表达式运算符(Expression operators)

- 算术运算符  
+ - \* /
- 和并符  
&
- 比较运算符  
EQ NE LT LE GT GE
- 布尔运算符  
NOT AND OR

VANTAGE Plant Design

## 表达式说明



- ▼ 表达式可以嵌套
  - ▼ 运算符前后必须加空格
  - ▼ 表达式前后的类型必须一致
- ```
!X = 64
!Y = 32
!Z = !X + !Y      !Z = 96
!A = 'Hello '
!B = 'World'
!AB = !A + !B      !AB = 'Hello World'
!XY = !X & !Y      !XY = '6432'
```

VANTAGE Plant Design

## 方法-实数对象(Real Object)



- ▼ 开方 Sqrt()
- ▼ 乘方 Power(REAL)
- ▼ 取整 INT()

VANTAGE Plant Design

## 定义函数




- ▼ 函数定义成全局变量

```
define function !!Area( !Length is REAL, !Width is REAL ) is REAL
    !Area = !Length * !Width
    return !Area
Endfunction
```

\$\*函数!!Area有两个参数一个返回值

- ▼ 文件命名和存放位置



- 文件名必须与函数名一样，后缀是小写的pmlfnc

VANTAGE Plant Design

## 定义函数



### ▼ 函数定义成全局变量

```
define function !!Area( !Length is REAL, !Width is REAL,!area is Real)
    !Area = !Length * !Width
Endfunction
```

VANTAGE Plant Design

## 函数调用



### ▼ 将新建函数加到索引文件Pml.index中

Pml Rehash All

### ▼ 调用函数并且得到返回值

```
!MyArea = !!Area(6,4)
Q var !MyArea
<Real> 24
```

### ▼ 练习-新建函数计算圆的面积,测试函数

```
define function !!circleArea( !radius is REAL) is REAL
    !Area = PI * !Radius.Power(2)
    return !Area
Endfunction
```

VANTAGE Plant Design

陈 鹏

## PML通用功能-注释



### ▼ 单行注释

--This is a new-style PML comment

-----

\$\* The following lines calculate the new angle

!Z = !X + !Y \$\* We are assuming both !X and !Y are REAL

### ▼ 段落注释

\$(

skip if (!X EQ !Y)

...

\$)

VANTAGE Plant Design

## PML通用功能-打印等



### ▼ 打印

\$P This text will be output to the screen.

### ▼ 换行

\$P This is an example of a much longer message \$  
that will be output to the screen

### ▼ 退出程序

if ( count EQ 0 ) then

return

endif

VANTAGE Plant Design



## 逻辑控制(Control Logic)



### ▼ 四种逻辑控制结构

- 条件判断语句 `If...else...endif`
- 循环语句 `Do ...enddo`
- 跳转 `Label Goto label`
- 错误处理 `Handle...Endhandle`

VANTAGE Plant Design

## 条件判断语句(If Construct)



### ▼ 判断表达式中必须是布尔值

```
!Type = Type
!OwnType = Type of Owner
IF (!Type eq 'BRAN') THEN
    $P CE is Branch.
ELSEIF (!OwnType eq 'BRAN') THEN
    $P CE is Branch member.
ELSE
    $P CE is $!Type,Pls select Branch.
ENDIF
```

### ▼ Elseif, else都是可选项

VANTAGE Plant Design

## 判断元素是否有名字



```
▼ 判断元素是否有名字
!Name = Name
!FullName = FullName
If(!Name EQ !FullName) then
    $p Named.
Else
    $p No name.
Endif
```

VANTAGE Plant Design

## 判断常用例程



```
▼ 判断变量是否定义
If(UnDefine(!x)) then
    $p Variables $!x not define.
endif

▼ 判断变量值是否存在
!x =real()

用函数判断                用方法判断
If(Unset(!x)) then...      if(!x.Unset()) then...
If(Set(!x)) then...         if(!x.Set()) then...
```

VANTAGE Plant Design

## 练习-条件判断



### ▼ 判断当前元素类型(Type), 添加当前元素和与之连接的元素

- 如果是'BRAN', 加入头尾连接的元素add Href Tref
- 如果是'NOZZ', 加入连接的管道add Cref
- 打印当前元素类型

```
!type = type
add ce
Auto ce
if(!type eq 'BRAN') then
  add href tref
elseif(!type eq 'NOZZ') then
  add cref
endif
$P Current type = $!type
```

VANTAGE Plant Design

## 循环(Do loops)



### ▼ 循环赋值

```
!Total = 0
Do !x From 1 To 100 By 1
  !Total = !Total + !x
Enddo
```

### ▼ 可以忽略的选项

- 如果循环起始为1, From选项可以忽略
- 如果步长为1, By选项可以忽略

VANTAGE Plant Design

## 中断循环 Break



### ▼ 中断循环 Break

```
!Total = 0
Do !x To 100
    !Total = !Total + !x
    If(!Total gt 500) then
        Break                $*或者Break if(!Total gt 500)
    Endif
Enddo
```

VANTAGE Plant Design

## 跳过循环(Skip)



### ▼ 用skip 跳过奇数

```
Do !x To 100
    If(Int(!x / 2) NE (!x / 2)) then
        Skip                $*或者Skip If(Int(!x / 2) NE (!x / 2))
    Endif
    !Total = !Total + !x
Enddo
```

VANTAGE Plant Design

## 跳转(Jump)



- ▼ 用golabel可以跳转到标记行,不限制前后顺序

Label /Start

...

GoLabel /Start

- ▼ Label名最长 16 个字符, 不包括 '/'

- ▼ 不允许跳转到Do循环中

golabel /illegal

!Total = 0

do !x from 1 to 5

!Total = !Total + !x

label /illegal

enddo

VANTAGE Plant Design

## 错误提示



- ▼ 测试程序

Next

\$p OK

- ▼ 可能出现的错误

(2,113) List exhausted

2 表示错误出现在PDMS中的哪个模块

113 是错误代码

- ▼ 出现错误通常有三种结果

- 出现一个警告框,用户必须确认
- 输出一个错误信息
- 从当前运行的程序中退出

VANTAGE Plant Design

## 错误处理(Error Handling)



### ▼ 处理特定错误，使程序继续执行

```
Next
Handle (2,113)
    $p Last element.
EndHandle
$p OK
```

### ▼ 处理任何可能的错误

```
Next
Handle Any
EndHandle
$p OK
```

VANTAGE Plant Design

## 处理多个错误



### ▼ 处理多个错误

```
Next
Handle (2,113)
    $p Last element.
ElseHandle (2,46)
    $p ...
EndHandle
$p OK
```

VANTAGE Plant Design

## 没有错误



### ▼ 处理没有错误

```
!noteM = !sheetname + '/LoadTable'  
$!noteM  
handle any  
    new note $!noteM  
elsehandle none  
    delete note  
    new note $!noteM  
endhandle
```

VANTAGE Plant Design

## 数组(Array)



### ▼ 从字符串创建数组

```
!Str = 'Benz,Bmw,Audi'  
!BestCar = !Str.Split(',')  
Q var !BestCar  
<ARRAY>  
[1] <STRING> 'Benz'  
[2] <STRING> 'Bmw'  
[3] <STRING> 'Audi'
```

\$\*数组元素，索引号

### ▼ 逐个元素添加

```
!BestCar[5] = 'Cadillac'
```

VANTAGE Plant Design

## 方法-数组对象(Array Object)



|          |                               |
|----------|-------------------------------|
| ▼ 数组元素数量 | Size()                        |
| ▼ 数组最大宽度 | Width()                       |
| ▼ 追加元素   | Append(value)                 |
| ▼ 追加数组   | AppendArray(Array)            |
| ▼ 删除全部元素 | Clear()                       |
| ▼ 删除单个元素 | !MyArray[N].Delete()          |
| ▼ 删除数组   | Delete()                      |
| ▼ 压缩数组   | Compress()                    |
| ▼ 排序     | Sort()                        |
| ▼ 倒序     | Invert()                      |
| ▼ 搜索     | Find(value), FindFirst(value) |
| ▼ 和并重复项  | Unique()                      |

VANTAGE Plant Design

## 数组方法实例



- ▼ 数组追加，数组必须存在,否则提前声明  
`!BestCar.Append('Lincoln')`
- ▼ 声明数组  
`!BestCar = array()`

VANTAGE Plant Design



## 数组循环



### ▼ 赋值循环Do value

```
do !Name values !BestCar $*将 !BestCar中元素逐个赋给!Name
  $p Array element is $!Name
Enddo
```

- 自动跳过空的数组元素

### ▼ 索引循环Do indices

```
do !n indices !BestCar $*将 !BestCar中索引号逐个赋给!n
  !Car = !BestCar[!n]
  $p Array element $!n is $!Car
Enddo
```

### ▼ 练习-测试数组循环

VANTAGE Plant Design

## PDMS中生成数组(Collections)



### ▼ 收集指定类型的Pdms元素

```
Var !PipeComps Collect All Pipe For CE
```

- 生成的!PipeComps是数组，保存的是元素的参考号
- 不使用For CE则是从整个MDB中提取

### ▼ 类型用法示例

|                      |             |
|----------------------|-------------|
| ALL                  | 所有元素        |
| ALL Pipe             | 所有的Pipe     |
| ALL BRANCH MEMBERS   | 所有管件，包括Tube |
| ITEMS OF EQUI /D1201 | 设备中的所有基本体   |

VANTAGE Plant Design

## 继续...



### ▼ 限制条件

Var !Elbows Coll All Elbow With (Abor gt 100) for CE

With后面是一个表达式

### ▼ 限制区域

Var !Elbows Coll All Elbow Within W0N0U0 to W2000N2000U2000

Var !Elbows Coll All Elbow Exclusive Within Volume /D1201 1500

Exclusive表示完全包含在空间内的元素才被选中

### ▼ 追加

Var !Elbows Append Coll all Bend for CE

VANTAGE Plant Design

## 练习



### ▼ 选择一个设备,将所有Nozzle连接的管道加入显示

- 加入当前设备,居中显示
- Nozzle的Cref属性记录连接的管道名称
- 通过参考号定位到Nozzle
- 加入显示Add Cref

Add ce

Auto ce

Var !nozzles Coll all nozzle for ce

Do !nozzle value !nozzles

\$!nozzle

Add cref

EndDo

VANTAGE Plant Design

## 求值(Evaluate)



### ▼ 通过参考号提取属性值

```
Var !Pipes Coll all Pipe for ce      $*得到参考号
Var !Names Eval name for all from !Pipes  $*提取Name属性
Q var !names
```

### ▼ 练习-将Zone中所有的Pipe名称和等级打印到命令行

– 使用Do循环将Name和Pspec输出到一行

```
Var !Pipes Coll all Pipe for ce      $*得到参考号
Var !Names Eval name for all from !Pipes  $*提取Name属性
Var !Pspecs eval pspec for all from !pipes
Do !n indices !names
    $p $!names[$!n] $!pspecs[$!n]
enddo
```

VANTAGE Plant Design

## 排列字符串Compose



### ▼ 按照宽度和对齐方式排列字符串,返回数组

```
!a = 'ABCDEFGF'
!b = 'DEF'
var !output compose '$!a' width 5 L SPACE 2 '$!b' width 5 R
Q var !output
```

### ▼ 输出数组的第一个元素

```
$p $!output[1]
```

VANTAGE Plant Design

## 练习



### ▼ 完善前一个练习，排列Name和Pspec

```
Var !Pipes Coll all Pipe for ce      $*得到参考号
Var !Names Eval name for all from !Pipes  $*提取Name属性
Var !Pspecs eval pspec for all from !pipes
Do !n indices !names
  var !out compose '$!names[$!n]' wid 15 space 2 '$!pspecs[$!n]' wid 15
  $p $!out[1]
enddo
```

VANTAGE Plant Design

## 练习



### ▼ 继续完善前一个练习，自动设置宽度Width()

```
Var !Pipes Coll all Pipe for ce      $*得到参考号
Var !Names Eval name for all from !Pipes  $*提取Name属性
Var !Pspecs eval pspec for all from !pipes
!wid1 = !names.width()
!wid2 = !pspecs.width()
Do !n indices !names
  var !out compose '$!names[$!n]' wid $!wid1 space 2 $
  '$!pspecs[$!n]' wid $!wid2
  $p $!out[1]
enddo
```

VANTAGE Plant Design




## 对话框(Form)



## 系统对话框






- ▼ 警告对象(Alert Objects)!!Alert是系统定义的对象
  - 三种无返回值的警告对象
  - 三种带返回值的警告对象



## 警告Alert Objects



- ▼ 三种无返回值的警告对象(Alert Object)
  - !!Alert.Error( 'You cannot do this!' )
  - !!Alert.Message( 'Saving your data now!' )
  - !!Alert.Warning( 'Do not press this button again!' )
- ▼ 缺省情况，警告窗口出现在光标附近，可用X，Y值指定其在屏幕上的位置,整个屏幕左上角是0,0,右下角是1,1
  - !!Alert.Error( 'You cannot do this!' , 0.25, 0.1)




VANTAGE Plant Design

## 确认-Confirm Alert Objects



- ▼ 三种带返回值的警告对象(Alert Object)
  - confirm, question 和 input
- ▼ 确认(Confirm),返回值是 'YES'或 'NO'
  - !Answer = !!Alert.Confirm( 'Are you sure!' )



VANTAGE Plant Design

## 练习



▼ 确认是否在管道上生成弯头，YES则生成，NO则退出

- 生成弯头New Elbow choose rtext
- 连接Conn

```
!Answer = !!Alert.Confirm('Creat Elbow?')
```

```
If(!answer eq 'YES') then
```

```
    New Elbow choose rtext
```

```
    conn
```

```
endif
```


VANTAGE Plant Design

## 询问-Question Alert Objects



▼ 询问(Question),返回值是 'YES' , 'NO'或 'CANCEL'

```
!Answer = !!Alert.Question( 'OK to delete Site?' )
```



VANTAGE Plant Design


## 输入-Input Alert Objects



### ▼ 输入(Input)

```
!Answer = !!Alert.Input( 'Enter Width of Floor','10' )
```

- 第一个值是输入提示，第二个值是缺省值
- Input alerts 返回一个字符串值



VANTAGE Plant Design

## 练习



### ▼ 完善上一个练习

- 如果生成弯头,询问出口方向和距离
- 改变方向Dir U
- 改变距离Dist 1000

```
!Answer = !!Alert.Confirm('Creat Elbow?')
```

```
If(!answer eq 'YES') then
```

```
  New Elbow choose rtext
```

```
  conn
```

```
  !dir = !!Alert.Input('Direction?', 'N' )
```

```
  dir $!dir
```

```
  !dist = !!Alert.Input('Distance?', '1000' )
```

```
  dist $!dist
```

```
endif
```

VANTAGE Plant Design




## 一个简单的对话框(Form)



- ▼ 对话框是全局变量的对象

```
setup form !!hello
  Title 'Hello'
  paragraph .Message text 'Hello world'
  button .bye 'Goodbye' OK
exit
```



- ▼ 控件(Gadget)

|                    |          |
|--------------------|----------|
| Paragraph .Message | 文字控件和控件名 |
| Button .bye        | 按钮控件和控件名 |

- ▼ 对话框控制属性OK


内置属性,关闭对话框

VANTAGE Plant Design

## 对话框调用



- ▼ 对话框存放位置



- ▼ 将新建对话框加到索引文件Pml.index中,在命令行键入

Pml Rehash All

- ▼ 显示对话框

Show !!hello

- ▼ 命令行关闭对话框

Hide !!hello

VANTAGE Plant Design

## 调试对话框



### ▼ 调试对话框

- 对话框显示后,定义文件被加载到内存中,下一次直接从内存中加载对话框定义内容
- 对话框的定义文件被修改后,需要重新加载  
Pml reload form !!hello  
Show !!hello

VANTAGE Plant Design

## 方法-对话框对象 (Form Object)



- ▼ 显示状态                      shown()
- ▼ 判断命令行窗口的显示状态  
!shown = !!CADCETH.shown()  
if(not !shown) then  
    show !!CADCETH  
endif  
var !date clock date  
var !time clock time  
\$p Begin report at \$!date \$!time

VANTAGE Plant Design

## 缺省构造方法(Constructor method)



### ▼ 增加一个输入框和方法

```
setup form !!hello
```

```
Title 'Hello'
```

```
paragraph .Message text 'Hello world'
```

```
text .input 'Enter text' width 10 is string
```

```
button .bye 'Goodbye' OK
```

```
exit
```

```
Define method .hello()
```


```
!this.input.callback = !this.doinput()
```

```
endmethod
```

```
Define method .doinput()
```

```
!this.message.val = !this.input.val
```

```
endmethod
```



VANTAGE Plant Design

## 缺省构造方法



### ▼ 缺省构造方法

- 与Form同名,不带参数
- 在调用对话框的时候自动执行
- 一般用于设置控件的缺省值和回调方法
- 再一次显示对话框时, 缺省构造方法不再执行, 因为对话框的定义内容已经全部加载到内存中,
- Kill !!hello从内存中清除已经加载的定义内容

VANTAGE Plant Design

## 控件的响应操作(Callbacks)




- ▼ 控件的callback成员用于指定控件的响应操作
- ▼ 控件的响应可以是以下的操作
  - 显示另一个对话框 form  
`!this.show.Callback= 'show !!hello'`
  - 直接执行一条命令command  
`!this.remove.Callback= 'remove all'`
  - 调用一个函数function  
`!this.area.Callback= '!area = !!area(!length,!width)'`
  - 调用一个方法method  
`!this.apply.Callback= '!this.Apply()'`

VANTAGE Plant Design


## 对话框布局 (Form Layout)



- ▼ 对话框有一个隐含的定位网格，原点在Form的左上角



- ▼ 每个控件(Gadget) 有四个隐含变量可以用于参考定位  
XMIN, XMAX, YMIN, YMAX



VANTAGE Plant Design

## 控件定位



### ▼ 自动定位(Auto-placement)

- **PATH Down** 下一个控件在前一个控件的下方
- **PATH up, PATH Right**(缺省), **PATH Left**

### ▼ 相对前一个控件定位(Relative)

```

setup form !!hello
Title 'Hello'
paragraph .Message text 'Hello world'
text .input 'Enter text' at x0 ymax width 10 is string
button .bye 'Goodbye' at x0 ymax OK
exit
    
```



### ▼ 相对指定控件定位

```

button .bye 'Goodbye' At xmax.input ymax.input OK
    
```


VANTAGE Plant Design

## 控件自动定位举例



```

button .But1          $* default placer
PATH down
HALIGN centre
VDIST 2
paragraph .Par2 width 3 height 2  $* auto-placed
toggle .Tog3             $* auto-placed
PATH right
HDIST 3.0
VALIGN bottom
list .Lis4 width 2 height 3      $* auto-placed
PATH up
HALIGN right
paragraph .Par5 width 3 height 3  $* auto-placed
    
```



VANTAGE Plant Design


## 对话框程序-Copy&Rename



▼ 实现拷贝同时修改名字

▼ 用到的命令

NEW type newname COPY oldname REN oldname newname By n 100



VANTAGE Plant Design

## 对话框定义文件



▼ 对话框定义文件

```
setup form !!mycopy
  title 'Copy and rename'
  text .from 'From' at x0 width 20 is string
  path right
  button .ce 'CE '
  text .to 'To ' at x0 ymax + 0.3 width 20 is string
  text .by 'By ' at x0 ymax + 0.3 width 10 is string
  button .copy 'Copy ' at x2 ymax + 0.3
  button .cancel 'Cancel' at x23 Cancel
exit
```

VANTAGE Plant Design

## 缺省构造方法



### ▼ 缺省构造方法

```
Define method .mycopy()  
  !this.ce.callback = '!this.ce()'  
  !this.copy.callback = '!this.copy()'  
  !this.by.val = 'N0'  
Endmethod
```

### ▼ 得到CE的名字

```
Define method .ce()  
  !this.from.val = name  
endmethod
```

VANTAGE Plant Design

## 主程序



### ▼ 首先判断控件的值

```
Define method .copy()  
  !from = !this.form.val  
  $!from  
  handle any  
    !!alert.message('$!from not existed.')  
    return  
  endhandle  
  !type = type  
  !to = !this.to.val  
  !by = !this.by.val  
  new $!type $!to copy $!from ren $!from $!to by $!by  
endmethod
```

VANTAGE Plant Design

## 控件的内置方法



### ▼ 控件的内置方法(Method)和成员(Member)

- 变灰一个控件  
`!this.apply.Active = FALSE`
- 聚焦控件  
`!this.input.SetFocus()`
- 编辑框只读  
`!this.input.seteditable(false)`
- 清除列表框的值  
`!this.list.clear()`
- 改变按钮上的文字  
`!this.apply.tag = ' Add '`
- 改变按钮的背景颜色  
`!this.apply.backgroud = 2`

VANTAGE Plant Design



## 复杂对话框程序






## 添加指定空间内的设计元素



### ▼ 添加指定空间内的设计元素



### ▼ 程序中用到的命令

Var !Comps Append Coll All pipe Within Volume /D1201 1500

VANTAGE Plant Design

## 控件定义(Gadget)



### ▼ 文本框Paragraph,对于Paragraph,at必须在text前面

paragraph .cname at x0 ymax text 'No name' width 20 height 1

### ▼ 编辑框Text

text .input 'Enter text' at x0 ymax width 10 is string

### ▼ 多选框Toggle

toggle .bran 'Branch'

toggle .equi 'Equipment'


### ▼ 框架Frame

Frame .frame1 'Include' at x0 ymax + 0.2

toggle .bran 'Branch'

toggle .equi 'Equipment'

Exit



VANTAGE Plant Design

## 继续...



### ▼ 按钮Button

```
button .ce 'CE' at x0 ymax tooltip 'Select pipe'
button .ok 'OK' at x0 ymax OK          $*关闭对话框,保留控件值
button .apply 'Apply' at x0 ymax Apply $*不关闭对话框
button .cancel 'Cancel' at x0 ymax Cancel $*关闭对话框,不保留控件值
```

### ▼ 图标按钮,png图像文件必须在Pmlib搜索路径内

```
!icon = !!pml.getpathname('modmodeon16.png')
button .pick pixmap /$!icon at x0 ymax
```

### ▼ 列表框List

```
list .list 'Drawlist' at x0 ymax width 18 height 5
```

### ▼ 练习-组合上述控件,生成新的对话框!!Addvolumn,比较Ok,Cancel的不同

VANTAGE Plant Design

## 设置控件初始值



### ▼ 缺省构造方法设置控件初始值

```
define method .addvolumn()
  !this.bran.val = true
  !this.input.val = '1500'
endmethod
```

VANTAGE Plant Design

## 按钮的应用实例



- ▼ 新建方法,将当前元素的名字显示在文本框中

```
define method .ce()  
  !this.cename.val = fullname  
endmethod
```

- ▼ 在缺省构造方法中指定控件的响应操作

```
define method .addvolumn()  
  ...  
  !this.ce.callback = '!this.ce()'  
endmethod
```

VANTAGE Plant Design

## 对话框控制属性-Apply



- ▼ 新建方法,调用控件的值

```
define method .apply()  
  !dist = !this.input.val  
  !name = !this.cename.val  
  $p CE is $!name ,Dist = $!dist  
endmethod
```

- ▼ 在缺省构造方法中指定Apply按钮的响应操作

```
define method .addvolumn()  
  ...  
  !this.apply.callback = '!this.apply()'  $*不关闭对话框  
endmethod
```

VANTAGE Plant Design

## 多选框状态判断



- ▼ 多选框(toggle)的状态通过.val成员判断  
`!this.bran.val = true` \$\*表示选中

VANTAGE Plant Design

## 多选框的应用实例



- ▼ 完善Apply方法

```
define method .apply()
  !name = !this.cename.val
  !dist = !this.input.val
  !elements = array()
  if(!this.bran.val) then
    var !elements append coll all bran Within Volume
    $!name $!dist
  endif
  if(!this.equi.val) then ...
  var !names eval name for all from !elements
  do !element value !elements
    add $!element
  enddo
endmethod
```

VANTAGE Plant Design

## 列表框应用实例



- ▼ 列表框用数组赋值,dtext表示display text

```
!this.list.dtext = !names
```

- ▼ 从列表框获取值

```
!name = !this.list.selection()
```

- ▼ 新建方法,提取List中的名字

```
define method .doselection()
```

```
!name = !this.list.selection()
```

```
$!name
```

```
endmethod
```

- ▼ 在缺省构造方法中指定列表框的响应操作

```
define method .addvolumn()
```

```
...
```

```
!this.list.callback = '!this.doselection()'
```

```
endmethod
```

VANTAGE Plant Design

## View控件



- ▼ 四种View控件

- 输入输出显示窗口 Alpha view
- Plot图片显示窗口 Plot view
- Design显示窗口 Volume view
- Draft显示窗口 Area View

VANTAGE Plant Design

## Alpha Views



### ▼ 自定义输入输出对话框

Setup Form !!alphaview

title ' Input & Output'

view .Input ALPHA hei 10 width 40

channel REQUESTS


channel COMMANDS

Exit

button .apply 'Apply' at x0 ymax Apply

button .Dismiss 'Cancel' at Xmax form-size Cancel

Exit



VANTAGE Plant Design

## Plot View



### ▼ 显示Plot图片

setup form !!plotview

View .view1 plot width 41 hei 23


exit

define method .plotview()

!this.view1.borders = false

!this.view1.add('c:/ladder.plt')

endmethod




VANTAGE Plant Design

## Column View



### ▼ Design的显示窗口

```
setup form !!poptest
view .vol volume width 50 hei 9
exit
path down
button .press 'swap popup' call '!this.popsnap()'
menu .popmenu
  add 'hello' ''
  add 'world' ''
exit
define method .popsnap()
  !this.vol.popup = !this.popmenu
endmethod
```



VANTAGE Plant Design



## 文件处理(Handling Files)



## 文件处理(Handling Files)



### ▼ 读写文件需要使用FILE对象

```
!Input = object FILE('%pdmsexex%\abc.txt')
!Lines = !Input.ReadFile()  $* ReadFile将文件内容写到字符串数组中
!ResultArray = ARRAY()    $* 声明新数组
do !Line VALUES !Lines
    !Column1 = !Line.Part(1)
    !ResultArray.Append( !Column1)
Enddo
```

```
!Output = object FILE('%pdmsexex%\def.txt')
```

```
!Output.WriteFile('WRITE', !ResultArray)  $* WriteFile将数组写到文件
```

### ▼ ReadFile()方法可以自动打开(Open)和关闭(Close)文件

### ▼ 写文件还有覆盖模式'OVERWRITE'和追加模式'APPEND'

VANTAGE Plant Design

## 练习-文件处理



### ▼ 将一个文本文件的内容显示在命令行中

```
!Input = object FILE('d:\abc.txt')
!Lines = !Input.ReadFile()  $* ReadFile将文件内容写到字符串数组中
do !Line VALUES !Lines
    $p $!Line
Enddo
```

### ▼ 将Pipe名称写入到一个文本文件中

```
Var !Pipes Coll all Pipe for ce          $*得到参考号
Var !Names Eval name for all from !Pipes  $*提取Name属性
!Output = object FILE('%pdmsexex%\def.txt')
!Output.WriteFile('overWRITE', !names)    $* 将数组写到文件
```

VANTAGE Plant Design



## 选择文件的标准函数




### ▼ 函数定义

define function

```
!!fileBrowser(!directory is STRING,  
              !seedFile is STRING,  
              !title is STRING,  
              !existFlag is BOOLEAN,  
              !callback is STRING)
```

### ▼ 使用实例

```
!!filebrowser('%pdmsuser%', '*',  
             'Text File', true,  
             '!filename = !!filebrowser.file.name()'  
             q var !filename')
```



VANTAGE Plant Design

## 实例应用



### ▼ 调用标准函数

setup form !!findfile

title 'Find file'

text .filename 'File name' at x1 ymax+0.3 width 35 is string

button .browser 'Browser'

button .ok ' OK ' at x1 ymax+0.3 OK

button .cancel 'Cancel' at x10 cancel


exit

define method .findfile()

!this.browser.callback =

```
!!filebrowser('%pdmsuser%', '*', "", true, !!findfile.filename.val  
             = !!filebrowser.file.name())
```

endmethod



VANTAGE Plant Design

## 练习-显示文件



- ▼ File Browser选择文件，将内容显示在命令行中

```
define method .findfile()
!this.browser.callback =
  [!!filebrowser('%pdmsuser%', '*', '', true, '!!findfile.filename
.val = !!filebrowser.file.name()')]
!this.ok.callback = '!this.display()'
endmethod
Define method .display()
!fname = !this.filename.val
!output = object file('$!fname')
!lines = !output.readfile()
q var !lines
endmethod
```

VANTAGE Plant Design



## 菜单(Menu)



## 菜单条



### ▼ Design系统菜单

– ..\AVEVA\Pdms11.5\pdmsui\DES\GEN\FSYSTEM

### ▼ 菜单条Bar

bar

add |Design| \_sysDes

add |Display| \_sysView

add |Query| \_sysQry

add |Settings| \_sysSet

add |Utilities| \_sysUtil

\$M /%Pdmsdfits%/usermenu/designmenu BAR \$\*调用自定义菜单条

handle ANY

endhandle

VANTAGE Plant Design

## 调用自定义菜单



### ▼ 调用自定义菜单，640行左右

-- Load Menus for Application

\$M/%Pdmsdfits%/usermenu/designmenu MENU \$\*调用自定义菜单

handle ANY

endhandle

### ▼ 调用自定义快捷按钮,700行左右

button .sysCla toggle ...

\$M/%Pdmsdfits%/usermenu/designmenu TOOLBAR

handle ANY

endhandle

VANTAGE Plant Design

## 自定义Design菜单



- ▼ 生成自定义菜单文件
  - %Pdmsdflts%\usermenu\designmenu

VANTAGE Plant Design

## 自定义菜单文件内容



- ▼ 文件内容示例

```
golabel /$1
Return
label /BAR
    add 'User' _UserTools          $* 定义菜单条
return
label /MENU
    menu _UserTools
        add 'Hello...' 'show !!hello'    $* 定义菜单项
    exit
Return
label /TOOLBAR
    !icon = !!pml.getpathname('modmodeon16.png')
    button .hello at xmax tooltip 'Hello' pixmap /$!icon width 16 height
    16 callback 'show !!hello'
return
```

VANTAGE Plant Design

## 自定义Draft菜单



- ▼ Draft系统菜单
  - ..\AVEVA\Pdms11.5\pdmsui\dra\gen\fsystem
- ▼ 生成自定义菜单文件
  - ..\AVEVA\Pdms11.5\pmlib\usermenu\draftmenu

VANTAGE Plant Design

## 自定义Catview菜单



- ▼ Catview系统菜单
  - ..\AVEVA\Catview11.3\Cat\Gen\Fsystem
- ▼ 生成自定义菜单文件
  - ..\AVEVA\Catview11.3\pmlib\usermenu\catmenu

VANTAGE Plant Design

**AVEVA™**